
Gambar 1. Uno Atmega 328P+ WiFi ESP8266 

https://robotdyn.com/uno-wifi-r3-atmega328p-esp8266-32mb-flash-usb-ttl-ch340g-micro-usb.html 

Pendahuluan. 
Arduino Ini adalah versi kustom dari ARDUINO UNO R3 klasik. Memiliki Integrasi penuh antara 

mikrokontroler Atmel ATmega328P dan IC Wi-Fi ESP8266 dengan memori flash 32 MB, dan 

konverter USB-TTL CH340G yang berada dalam satu papan (Board), dimana semua modul dapat 

bekerja sama atau berdiri sendiri. 

Memiliki tempat sakelar untuk memilihi 8 posisi mode 

 

 

Gambar 2. Switch Mode 

 

 

Tabel 1. Pemilihan Mode dan Status Switch 

 

 

 



 

Gambar 3. Tombol ESP Reboot 

 

Setelah memilih mode yang digunakan, jika anda menggunakan ESP8266 sangat penting untuk menekan 

tombol “ESP-Reboot”.  



Studi Kasus: 
Misalnya kita ingin mengambil data Suhu dan Kelembaban dengan Menggunakan Sensor DHT11, dari 

DHT11 data diambil  diproses Atmega 328P kemudian hasilnya ditampilkan di LCD dan Dikirim melalui 

komunikasi Serial ke ESP8266. Dari ESP8266 data dikirimkan  ke database mysql,  

 

Sensor DHT11 

 

 

Gambar 4. Skema Pengkabelan Pada Arduino dan DHT11 

 



1. Memprogram Atmega328p 
Pertama Mikrokontroler kita setting untuk menghubungkan antara  CH340 connect to ATmega328 (upload 

sketch)  yaitu dengan menggeser posisi Switch 3 dan switch 4 ke posisi ON, sesuai table 1 diatas 

 

Gambar 5. Pemindahan Switch Mode 

 

Adapun Kode Programnya adalah sebagai berikut: 

//*************************** 

//*************************************************************** 

//Nama Sketch:robodyn1 

//Mengambil data dari DHT11 Mengirim ke Serial 2 dan 3 ESP8266 

//Taryanarx@gmail.com 

//https://ciwaruga.com 

//*************************************************************** 

#include <Wire.h> // Library komunikasi I2C  

#include <SoftwareSerial.h> 

#include <LiquidCrystal_I2C.h> // Library modul I2C LCD 

 

#include <dht.h> 

#define sensor 8 //Digital 8  

dht DHT; 

float humidity, temperature; 

SoftwareSerial ss(2,3); 

  

// default address 0x27  

// tipe LCD 16x2 (16,2) 

LiquidCrystal_I2C lcd = LiquidCrystal_I2C(0x27, 16, 2);  

  



 

void setup() { 

  lcd.init(); 

  lcd.backlight(); 

  lcd.clear(); 

  //         1234567890123456 

  lcd.print("  Allisa Kost ");  

  lcd.setCursor(0, 1);  

  lcd.print(" CIWARUGA.COM "); 

  Serial.begin(9600); 

  ss.begin(115200); 

  delay(5000); 

} 

  

void loop() { 

  DHT.read11(sensor); 

  humidity=DHT.humidity; 

  temperature=DHT.temperature; 

 

  lcd.clear();  

  lcd.print("Humidity= "); 

  lcd.print(humidity); 

  lcd.print("%"); 

  lcd.setCursor(0, 1);  

  lcd.print("Tempertr= "); 

  lcd.print(temperature);  

  lcd.print("C"); 

 

  //kirim data ke ESP melalui komunikasi serial  

  String sendToESP = ""; 

  sendToESP += humidity; 

  sendToESP += ";"; 

  sendToESP += temperature; 

  ss.println(sendToESP); 

  delay(1000); 

} 

 

Kemudian Jika Telah Berhasil mengupload Sketch ke Atmega maka akan ditampilkan hasil seperti 

Gambar 6. berikut di LCD monitor 



 

Gambar 6. Hasil Keluaran Pengukuran di LCD 

 

2.Memprogram ESP8266 
Berikutnya setelah tadi berhasil memprogram Atmega, Selanjutnya adalah  memprogram ESP8266 

terlebih dahulu pindahkan switch ke mode CH340 connect to ESP8266 (upload sketch) (switch 5, 6 dan 7 

ON). 

Board mikrokontroler ini compatible dengan NodeMCU, maka kita akan menggunakan board NodeMCU 

pada pemilihan board di software Arduino IDE-nya. 



 

Gambar 7. Memilih Board NodeMCU 

Kode Program ESP8266 Untuk Merima Data yang dikirimkan dari Atmega, kemudian dikirimkan ke 

Server 

//***************************^^^^^^^^^^^^^^^^^^^^^^^ 

//NamaFile:robodyn2 

//Menerima data melalui serial dari atmega ke esp8266 

//Taryanarx@gmail.com 

//https://ciwaruga.com 

//*************************************************** 

#include <ESP8266WiFi.h> 

#include <WiFiClient.h>  

#include <ESP8266WebServer.h> 

#include <ESP8266HTTPClient.h> 

 

const char *ssid     = "Bohay";   

const char *password = "51ngsabar"; 

 

 

float humidity = 0, temperature = 0; 

unsigned long currentMillis = 0, prevMillis = 0, intervalMillis = 2000; 

 

void setup() { 

  Serial.begin(115200); 

  connectToWifi(); 

} 

 

void connectToWifi(){ 

  WiFi.mode(WIFI_OFF);         

  delay(1000); 

  WiFi.mode(WIFI_STA);         



   

  WiFi.begin(ssid, password);  

  Serial.println(""); 

 

  Serial.print("Connecting"); 

  // Wait for connection 

  while (WiFi.status() != WL_CONNECTED) { 

    delay(500); 

    Serial.print("."); 

  } 

  Serial.println(""); 

  Serial.print("Connected to "); 

  Serial.println(ssid); 

  Serial.print("IP address: "); 

  Serial.println(WiFi.localIP()); 

} 

 

String splitString(String data, char separator, int index) 

{ 

    int found = 0; 

    int strIndex[] = { 0, -1 }; 

    int maxIndex = data.length() - 1; 

 

    for (int i = 0; i <= maxIndex && found <= index; i++) { 

        if (data.charAt(i) == separator || i == maxIndex) { 

            found++; 

            strIndex[0] = strIndex[1] + 1; 

            strIndex[1] = (i == maxIndex) ? i+1 : i; 

        } 

    } 

    return found > index ? data.substring(strIndex[0], strIndex[1]) : ""; 

} 

 

void kirimDataKeServer(){ 

  HTTPClient http; 

  String postData; 

  String url; 

  //Post Data 

  postData = "humidity="; 

  postData += humidity; 

  postData += "&temperature="; 

  postData += temperature; 

 

  //ipaddress server apache mysql 

  url="http://192.168.0.105/arduino/server.php"; 

   WiFiClient client; 

   http.begin(client, url); 

   http.addHeader("Content-Type", "application/x-www-form-urlencoded"); 

       

  int httpCode = http.POST(postData);   //Send the request 

  String payload = http.getString();    //Get the response payload 

       

  Serial.println(httpCode);   //Print HTTP return code 

  Serial.println(payload);    //Print request response payload 

       

  http.end(); 

} 



 

 

void loop() { 

  if(Serial.available()){ 

    String msg = ""; 

    while(Serial.available()){ 

      msg += char(Serial.read()); 

      delay(50); 

    } 

 

    humidity    = splitString(msg, ';', 0).toFloat(); 

    temperature = splitString(msg, ';', 1).toFloat(); 

  } 

   

  currentMillis = millis(); 

  if((currentMillis - prevMillis) >= intervalMillis){ 

    kirimDataKeServer(); 

    prevMillis = currentMillis; 

  } 

} 

 

Lakukan Kompilasi dan Upload sketch ke ESP8266, Jika berhasil maka pada layar bawah akan tampil 

informasi seperti  Gambar 8.  



 

Gambar 8. Upload Sketch ke ESP8266 Berhasil 

 

Sampai tahap sini proses Upload Sketch ke ESP8266 telah berhasil. 

  



 

3.Menghubungkan PIN TX dan RX ESP8266 
Langkah berikutnya adalah menghubungkan secara Serial Pin TX ESP ke PIN D2, dan Pin RX ESP ke PIN D3 

AtMega, Kemudian pindahkan switch ke mode Mega328+ESP8266, yaitu Switch 1 dan 2 ON 

 

 

Gambar 9. PIN ESP8266 

 

Panggil Kembali Program yang pertama yaitu robodyn1,  Kemudian Upload ke Arduino Uno, 

selanjutnya pada LCD akan ditampilkan data hasil pembacaan dan apabila dilihat di database, 

maka lognya akan terus bertambah, seperti berikut: 



 

Gambar 10. Log hasil Pengukuran 

 

Koneksi.php 

<?php 

    $servername = "localhost"; 

    $username   = "root"; 

    $password   = ""; 

    $dbname     = "arduino"; 

    $conn = new mysqli("$servername", "$username", "$password","$dbname"); 

    // Check connection 

    if ($conn -> connect_errno) { 

         echo "Failed to connect to MySQL: " . $conn -> connect_error; 

    exit(); 

} 

?> 

  



 

kirimkeserver.php 

<?php 

  include("koneksi.php"); 

  if(!empty($_POST)){ 

    $humidity = $_POST["humidity"]; 

    $temperature = $_POST["temperature"]; 

    $query = "INSERT INTO log (humidity, temperature) 

            VALUES ('".$humidity."', '".$temperature."')"; 

    if ($conn->query($query) === TRUE) { 

      echo "Berhasil menyimpan data ke table log"; 

    } else { 

      echo "Error: " . $sql . "<br>" . $conn->error; 

    } 

  } 

?> 

 

Tampillog.php 

<!DOCTYPE html> 

<html> 

<title>Menampilkan Data Suhu dan Kelembaban</title> 

<meta name="viewport" content="width=device-width, initial-scale=1"> 

<link rel="stylesheet" href="https://www.w3schools.com/w3css/4/w3.css"> 

<body> 

<div class="w3-container"> 

  <h1> <center>Data Hasil Pengukuran</h1> 

     <h3><center>Sensor DHT11</h3> 

 

  <table class="w3-table-all"> 

    <thead> 

      <tr class="w3-light-blue"> 

        <th>NO</th> 

        <th>KEGANTENGAN</th> 

        <th>KEUANGAN</th> 

  <th>WAKTU</th> 

      </tr> 

    </thead> 

<?php 

require ("koneksi.php"); 

$sql="select * from log order by waktu desc"; 

$hasil=mysqli_query($conn,$sql); 

$row=mysqli_fetch_row($hasil); 

$n=1; 

do 

{ 

list($id,$suhu,$kelembaban,$waktu)=$row; 

echo "<tr><td>$n</td><td>$suhu %</td><td>$kelembaban %</td><td>$waktu</td>"; 

$n++; 

} 

while($row=mysqli_fetch_row($hasil)); 

?>   

  </table> 

</div> 

</body> 

</html> 

 

 

 



Buat Tabel Log 

 

 

Penutup 
Ok Sekian dulu terimakasih, semoga membantu, dan semoga kita selalu diberi rizky yang cukup dan selalu 

sehat. 

 

Aamiin YRA 

 

 

 

 

 


